UDK 528.33:528.181
Stručni rad

PROGRAM ZA RJEŠAVANJE SISTEMA LINEARNIH JEDNAČINA METODOM CHOLESKOG

*Zdravko GALIĆ — Sarajevo**

1. UVOD

Za rješavanje sistema linearnih jednačina postoji dosta metoda, ali se zbog svoje ekonomičnosti Gaussov metod eliminacije najčešće koristi. S druge strane, korišćenje matričnog računa, zbog svoje preglednosti i lakšeg programiranja računara, našlo je svoju primjenu u interpretaciji raznih naučno-tehničkih problema. Kod rješavanja matričih jednačina, najsloženiji skup operacija je računanje inverzne matrice, za što postoji više postupaka. Ovdje će biti prikazan algoritam i program za rješavanje sitema linearnih jednačina u kome je korištena metoda Choleskog računanja inverzne matrice, koja se primjenjuje kod simetričnih matrica. Na osnovu ovog algoritma i njegovog dijagrama toka, moguće je sastaviti program za razne vrste i tipove računara.

2. RAČUNANJE INVERZNE MATRICE METODOM CHOLESKOG

Cholesky je uočio da se kvadratna simetrična matrica može rastaviti u proizvod dvije trouglaste matrice. Suština ovog postupka je u tome, da se, kada je A simetrična matrica reda »n«, odredi trouglasta matrica

$$R = \begin{bmatrix} r_{11} & r_{12} & r_{13} & \cdots\cdots & r_{1n} \\ 0 & r_{22} & r_{23} & \cdots\cdots & r_{2n} \\ \cdots & \cdots & \cdots & \cdots & \cdots \\ 0 & 0 & 0 & \cdots\cdots & r_{nn} \end{bmatrix}$$

i njena transponovana matrica R′ koje zadovoljavaju jednačinu

$$A = R'R \tag{1}$$

* Adresa autora: Zdravko Galić, dipl. inž. Građevinski fakultet Sarajevo, Hasana Brkića 24.

odnosno,

$$\begin{bmatrix} a_{11} & a_{12} & a_{13} & \dots & a_{1n} \\ a_{12} & a_{22} & a_{23} & \dots & a_{2n} \\ \dots & \dots & \dots & \dots & \dots \\ a_{1n} & a_{2n} & a_{nn} & \dots & a_{nn} \end{bmatrix} = \begin{bmatrix} r_{11} & 0 & 0 & \dots & 0 \\ r_{12} & r_{22} & 0 & \dots & 0 \\ \dots & \dots & \dots & \dots & \dots \\ r_{1n} & r_{2n} & r_{3n} & \dots & r_{nn} \end{bmatrix} \begin{bmatrix} r_{11} & r_{12} & r_{13} & \dots & r_{1n} \\ 0 & r_{22} & r_{23} & \dots & r_{2n} \\ \dots & \dots & \dots & \dots & \dots \\ 0 & 0 & 0 & \dots & r_{nn} \end{bmatrix} \tag{2}$$

Iz (2) se može napisati n(n + 1)/2 jednačina, iz kojih se mogu odrediti elementi r_{ij}, a zbog simetričnosti je dovoljno odrediti samo elemente matrice R. Tako se dobije:

$$r_{11} = \sqrt{a_{11}}$$

$$r_{ij} = \frac{a_{ij}}{r_{11}} \qquad \text{za } i = 1$$

$$r_{ij} = \left(a_{ij} - \sum_{k=1}^{i-1} r_{ki}\, r_{kj}/r_{ii}\right) \qquad \text{za } i < j \tag{3}$$

$$r_{ii} = \sqrt{a_{ii} - \sum_{k=1}^{i-1} r_{ki}{}^2} \qquad \text{za } j = i$$

pri čemu se za dijagonalne elemente r_{ii} uzima pozitivni kvadratni korijen Rješenje matrične jednačine

$$AY = B \tag{4}$$

pod uslovom da je matrica A regularna ($\|A\| \neq 0$), je

$$Y = XB \tag{5}$$

gdje je X, inverzna matrica, matrice A.

Za dobivanje inverzne matrice X polazi se od jednačine

$$AX = E \tag{6}$$

gdje je E jednačina matrica.

Ako matricu X napišemo kao vrstu vektor-kolona

$$X = [x_1, x_2, \dots x_n] \tag{7}$$

gdje je

$$x_i = [x_{i1}, x_{i2}, \dots, x_{in}]$$

onda se (6) može napisati:

$$AX = [Ax_1, Ax_2, \dots, Ax_n] = E \tag{8}$$

odakle se dobije niz sistema

$$\begin{aligned} Ax_1 &= e_1 \\ Ax_2 &= e_2 \\ &\dots\dots\dots \\ Ax_n &= e_n \end{aligned} \tag{9}$$

gde su

$$e_1 = [1\ 0 \dots 0],\ e_2 = [0\ 1 \dots 0], \dots\ e_n = [0\ 0 \dots 1]$$

U matričnoj algebri se dokazuje da, ako se pođe od matričine jednačine

$$AX = B \tag{10}$$

istu možemo metodom Choleskog svesti na oblik:

$$RX = r \tag{11}$$

Praktično, to znači da treba desno pored matrice A dodati kolonu (vektor slobodnih članova) B, i za tako proširenu matricu metodom Choleskog odrediti trouglastu matricu [Rr].

Napišimo jednačinu (11) u razvijenom obliku

$$\begin{aligned} r_{11} x_1 + r_{12} x_2 + \dots + r_{1n} x_n &= r_1 \\ r_{22} x_2 + \dots + r_2\ x_n &= r_2 \\ \dots\dots\dots\dots\dots\dots\dots\dots\dots\dots\dots\dots \\ r_{nn} x_n &= r_n. \end{aligned}$$

Kao i kod Gaussovog postupka, računanje se vrši od poslednje nepoznate, a u opštem slučaju je:

$$x_i = \left(r_i - \sum_{j=i+1}^{n} r_{ij} x_j / r_{ii} \right).$$

U sistemu (9) matrica A je ista, pa će i matrica R također biti ista za sve sisteme. Samo se sada umjesto jedne kolone B javljaju e_1, $e_2, \dots, e_n$, za koje se određuje n transformisanih vektora-kolona $r_i (i = 1, 2, \dots, n)$. Nakon toga se računaju vektori $x_i (i = n, n-1, n-2, \dots, 1)$, tj vrste inverzne matrice, počevši od poslednje.

3. DIJAGRAM TOKA I PROGRAM

Na ulazu se vrši učitavanje reda matrice — n, matrice A i vektora slobodnih članova B. Zatim se vrši generisanje matrice [AE], određivanje matrice [Rr] i računanje vektora x_i na opisani način. Kod računanja dijagonalnih elemenata matrice R može se dogoditi da potkorjena veličina u izrazu za r_{ii} bude negativna, pa veličine r_{ii} postaju imaginarne. Za samo računanje to ne


START
ULAZ: N,A,B
N1=N+1
M=2*N
I=1,N
J=N1,M
J=N+I
DA
NE
A(I,J)=0.
A(I,J)=1.
R(1,1)=(A(1,1))**0.5
J=2,M
R(1,J)<=A(1,J)/R(1,1)
I=2,N
J=I,M
L=I-1
DA
NE
I=J
R(I,J)=0.
R(I,J)=C.
K=1,L
K=1,L
R(I,J)<=R(I,J)+R(K,I)**2
R(I,J)<=R(I,J)+R(K,I)*R(K,J)
R(I,J)<=(A(I,J)-R(I,J))**0.5
R(I,J)<=(A(I,J)-R(I,J))/R(I,I)
1



predstavlja poteškoću, jer i svi ostali elementi r_{ij} iste vrste postaju imaginarni, a pošto se oni u daljem postupku množe međusobno u vidu skalarnog proizvoda, kao rezultati se dobijaju realni brojevi.

Međutim, kod izrade programa o ovome je potrebno voditi računa, i obično treba deklarisati elemente matrice R kao kompleksne promjenljive.

```
      COMPLEX R
      DIMENSION A(5,10),X(5,5),R(5,10),Y(5),B(5)
      READ(1,1)N
    1 FORMAT(I5)
      N1=N+1
      M=2*N
      DO 2 I=1,N
    2 READ(1,3)(A(I,J),J=1,N)
    3 FORMAT(5F5.2)
      DO 4 I=1,N
    4 READ(1,3)B(I)
      DO 6 I=1,N
      DO 6 J=N1,M
      IF(J.EQ.N+I)GO TO 5
      A(I,J)=0.
      GO TO 6
    5 A(I,J)=1.
    6 CONTINUE
      R(1,1)=SQRT(A(1,1))
      DO 7 J=2,M
    7 R(1,J)=A(1,J)/R(1,1)
      DO 10 I=2,N
      DO 10 J=I,M
      L=I-1
      IF(I.EQ.J)GO TO 9
      R(I,J)=0.
      DO 8 K=1,L
    8 R(I,J)=R(I,J)+R(K,I)*R(K,J)
      R(I,J)=(A(I,J)-R(I,J))/R(I,I)
      GO TO 10
    9 R(I,J)=0.
      DO 11 K=1,L
   11 R(I,J)=R(I,J)+R(K,I)**2
      R(I,J)=CSQRT(A(I,J)-R(I,J))
   10 CONTINUE
      DO 12 J=1,N
   12 X(N,J)=R(N,N+J)/R(N,N)
      N2=N-1
      DO 14 K=1,N2
      I=N-K
      I1=I+1
      SUM=0.
      DO 14 L=1,N
      DO 13 J=I1,N
   13 SUM=SUM+R(I,J)*X(J,L)
      X(I,L)=(R(I,N+L)-SUM)/R(I,I)
      SUM=0.
   14 CONTINUE
      DO 15 I=1,N
      Y(I)=0.
      DO 15 J=1,N
   15 Y(I)=Y(I)+X(I,J)*B(J)
      WRITE(3,20)
   20 FORMAT('1'///25X'MATRICA A I VEKTOR SLOBODNIH ČLANOVA B'//)
      DO 16 I=1,N
   16 WRITE(3,30)(A(I,J),J=1,N),B(I)
   30 FORMAT(22X5F6.2,7XF6.2/)
      WRITE(3,40)
   40 FORMAT(///25X'INVERZNA MATRICA X I VEKTOR RJEŠENJA Y'//)
      DO 17 I=1,N
   17 WRITE(3,30)(X(I,J),J=1,N),Y(I)
      STOP
      END
```

Za dobivanje vektora rješenja Y potrebno je izvršiti množenje inverzne matrice X i vektora slobodnih članova B. Na izlazu se vrši štampanje ulaznih podataka, inverzna matrica X i vektor rješenja Y.

Po ovom algoritmu kodiran je program u FORTRAN-u i za ilustraciju je riješena matrična jednačina:

$$\begin{bmatrix} 3.0 & -1.0 & 2.0 & 0.0 & 1.0 \\ -1.0 & 6.0 & 2.0 & 1.0 & 0.0 \\ 2.0 & 2.0 & 4.0 & -2.0 & -1.0 \\ 0.0 & 1.0 & -2.0 & 5.0 & 3.0 \\ -1.0 & 0.0 & -1.0 & 3.0 & 4.0 \end{bmatrix} \cdot \begin{bmatrix} y_1 \\ y_2 \\ y_3 \\ y_4 \\ y_5 \end{bmatrix} = \begin{bmatrix} 3.0 \\ 1.0 \\ -15,0 \\ 34.0 \\ 22.0 \end{bmatrix}$$

MATRICA A I VEKTOR SLOBODNIH ČLANOVA B

3.00	−1.00	2.00	0.0	1.00	3.00
−1.00	6.00	2.00	1.00	0.0	1.00
2.00	2.00	4.00	−2.00	−1.00	−15.00
0.0	1.00	−2.00	5.00	3.00	34.00
1.00	0.0	−1.00	3.00	4.00	22.00

INVERZNA MATRICA X I VEKTOR RJEŠENJA Y

9.36	5.50	−9.86	−3.93	−1.86	7.00
5.50	3.50	—6.00	−2.50	−1.00	3.00
−9.86	−6.00	10.86	4.43	1.86	−7.00
−3.93	−2.50	4.43	2.21	0.43	4.00
−1.86	−1.00	1.86	0.43	0.86	−1.00

LITERATURA:

[1] Anđelić T.: Matrice, Građevinska knjiga, Beograd, 1979.
[2] Brčić V.: Tehnika računanja , Građevinska knjiga, Beograd, 1961.
[3] Čubranić N.: Teorija pogrešaka sa računom izjednačenja, Tehnička knjiga, Zagreb, 1966.
[4] Parezanović N.: Računske mašine i programiranje — programski jezik FORTRAN IV, Beograd 1977.
[5] Pašalić S.: Računska tehnika i odabrane numeričke metode, Sarajevo, 1973.

REZIME

U radu je prikazan program za rješavanje sistema linearnih jednačina metodom Choleskog, pogodan za primjenu u geodeziji. Izložena je i sama metoda i date njene karakteristike.

ABSTRACT

In this paper is presented the algorithm for solution the system of linear aquations by Cholesky's method, which is fovourable to application in geodesy. The metod and its characters is also shown.

Primljeno: 1982-07-08